
まえがき

まえがき

本書は、SPARQL*1の初学者が実践力を身に着けることを目的に書かれた本です。入門的な内容だ

けではなく、実践的なクエリがまとまった資料はWeb上でもまだ少ないのが現状です。そこで弊研究

室の SPARQLer(SPARQLを操る者)の皆さんの協力の下、様々なクエリをかき集めました。集めら

れたクエリを難易度に応じて分類し、演習としてサンプルクエリを実行しながら読み進められるノック

集として構成しました。基本を学んだ後のステップアップのための「初級～上級」、そして SPARQL

を極める「職人級」まで用意しています。楽しんでいただければ幸いです。

本書で扱う SPARQLは、2013年にW3C勧告となった SPARQL 1.1*2の仕様を基に記述していま

す。しかしながら、実用上は特定の実装に依存する機能を使ったり、仕様に違反する挙動に対応したり

する必要を迫られることがあると思われます。現に、我々は本書で紹介するクエリを出し合う中で、そ

ういった場面に沢山遭遇しました。標準の仕様に準拠していないクエリをいくつか紹介していますが、

可能な限り注意書きを入れていますので、お読みになる際はその点もご確認ください。

DBpedia Japaneseを利用するサンプルクエリについては、2021-12-01版データセットでクエリの

実行を確認しています。

ライセンスについて

本書 (PDF版)は「クリエイティブ・コモンズ 表示 - 非営利 - 継承 4.0 国際 ライセンス*3」に基づ

いて利用することができます。

*1 https://www.w3.org/TR/sparql11-overview/
*2 https://www.w3.org/TR/2013/REC-sparql11-query-20130321/
*3 https://creativecommons.org/licenses/by-nc-sa/4.0/

第 1 章

ノックを始める前に
まずは準備運動から始めよう！

これから 50本ノックを始めるにあたり、本章では SPARQLクエリを実行する環境について説明し

ます。

1.1 公開 SPARQLエンドポイントを使う

Web 上には様々な Linked Open Data(LOD) データセットが公開されており、The Linked Open

Data Cloud*1ではそれらを可視化したグラフで把握することができます。全ての LODデータセット

が SPARQLエンドポイント提供しているとは限りませんが、利用可能なエンドポイントをいくつか見

つけることができます。

本書の 50本ノックでは、規模が大きく幅広いデータが存在する DBpedia Japaneseの SPARQLエ

ンドポイントを主に利用します。DBpediaとは、Wikipediaに記述された情報から構造化情報を抽出

してデータセットを生成するプロジェクトまたはそのデータセットのことです。DBpedia Japaneseは

日本語版Wikipediaを対象として生成されたデータセットです。Webブラウザからアクセス可能なイ

ンタフェースがありますので、お手元の PCに特別の準備は不要でノックを楽しむことができます。

それでは早速DBpedia Japanese(http://ja.dbpedia.org/)にアクセスしてみましょう (図 1.1)。

図 1.1 DBpedia Japaneseトップページ

ページ上部のナビゲーションバーから「SPARQL Endpoint」のリンクをクリックします。

*1 https://lod-cloud.net/

第 1章 ノックを始める前に

図 1.2 DBpedia Japanese SPARQLエディタ

図 1.2の SPARQLクエリエディタが表示されれば、これで準備は完了です。あとは「Query Text」の

テキストフォームに SPARQLクエリを入力し「Run Query」ボタンをクリックすればクエリが実行さ

れます。

なお、このWebインタフェースは SPARQLサーバの一つである OpenLink Virtuoso*2によって提

供される機能です。全ての SPARQLエンドポイントがこのようなWebインタフェースを提供してい

るわけではありません。

1.2 SPARQLクエリエディタ YASGUI

前節では DBpedia JapaneseのWebインタフェースの使い方を紹介しました。しかし、クエリを入

力し実行するための基本的な機能しかありませんので、より複雑なクエリを効率的に書くためには少々

不十分です。ここでは特定のエンドポイントに依存しないクエリエディタ YASGUIを紹介します。

YASGUI(Yet Another Sparql GUI)*3は、構文のバリデーションや入力補完など備えたクエリエ

ディタと、クエリ結果を様々なグラフで表示する可視化機能を提供します。特に構文のバリデーション

機能は SPARQL初学者にとって非常に有用ですので、ノックでは基本的に YASGUIを使うことをお

すすめします。他にも設定次第で様々な機能を使うことができますので、興味のある方はドキュメン

ト*4をご参照ください。

それでは使い方を簡単に説明します。Webブラウザ上で動作するツールですので、インストール作

業などは不要です (CORSプロキシや短縮 URL作成など一部の機能は別途サーバソフトウェアが必要

です)。公式サイトにて使用可能な状態で提供されているので、今回はこちらを使って説明します。

https://yasgui.triply.cc/にアクセスすると、次のようなエディタが表示されます（図 1.3）。

エディタ上部のテキストボックスにはエンドポイント URI を指定します。試しに DBpedia

*2 https://virtuoso.openlinksw.com/

http://vos.openlinksw.com/owiki/wiki/VOS/
*3 https://yasgui.triply.cc/
*4 https://triply.cc/docs/yasgui-api

図 1.3 YASGUI

Japanese のエンドポイント URI(https://ja.dbpedia.org/sparql) を入力してみます。あと

はクエリを入力した後、右上の �ボタンをクリックすればクエリが実行されます。また、 ボタンを

クリックすると、そのクエリを共有するためのリンクが生成されます。渾身の一本をみんなに自慢しま

しょう。

1.3 任意のデータに対して SPARQLクエリを実行する

ここまででは、特定の SPARQLエンドポイントに対してクエリを実行する方法を紹介しました。本

節では、自分で用意した RDFデータに対してクエリを実行する方法を紹介します。

一般的には、OpenLink VirtuosoやApache Jena*5などのRDFストアをインストールして SPARQL

サーバを構築することが必要です。しかしながら、このような環境構築で読者の皆さんを消耗させるこ

とは本書の意図するところではありません。

そこで RDFShape*6というツールを紹介します。このツールは RDFデータを検証するためのデモ

として作成されたようですが、RDFに関連する様々な操作を試すことができる非常に便利なツールで

す。個々の機能を紹介するだけでもう一冊本が書けそうなので、SPARQLの部分のみを取り上げます。

RDFShapeはローカルにインストールすることもできますが、こちらもWeb上で公開されているデモ

を利用することにします。

http://rdfshape.herokuapp.com/queryにアクセスします。

*5 https://jena.apache.org/
*6 RDFShape: RDF playground, http://rdfshape.herokuapp.com/

https://github.com/labra/rdfshape

第 1章 ノックを始める前に

図 1.4 RDFShape SPARQL

図 1.4 のような画面が表示されます。左側に RDF データを、右側に SPARQL クエリを入力し

「Query」をクリックすればクエリが実行されます。入力する RDFデータは、Turtleや TriG、JSON-

LD など主要なフォーマットはサポートされていますので、左下の「Data Format」から適切なもの

を選択してください。そのままテキストで入力する以外にも、アクセス可能な URIやファイルアップ

ロードで入力することも可能です。

次の図 1.5はクエリを実行した例です。結果は上部にテーブル形式で表示されます。

図 1.5 RDFShape SPARQLクエリ実行例

第 2 章

初級基本の構文や関数を確認しながら軽く打っていこう！

それではノックを始めましょう。各ノックに表示している Endpoint は、そのクエリ対象の

SPARQL エンドポイント URI を示します。「汎用」と表示されているものは、特定のエンドポイ

ントに依存しないクエリです。特定の SPARQLサーバの実装に依存クエリについては、その製品名称

(「OpenLink Virtuoso」など)を表示しています。

1本目 クラス IRIのリストを取得する

Endpoint 汎用

1 SELECT DISTINCT ?class
2 WHERE {
3 ?s a ?class .
4 }

そのままです。なお、キーワード「a」は rdf:typeと等価です (大文字小文字を区別します)。

2本目 グラフ IRIのリストを取得する

Endpoint 汎用

1 SELECT DISTINCT ?g
2 WHERE {
3 GRAPH ?g {
4 ?s ?p ?o .
5 }
6 }

名前付きグラフの IRIをすべて取得します。GRAPHキーワードを明示的に指定しない場合は、デ

フォルトグラフがクエリの対象となります (詳細は�35本目)。

3本目 ASKでクエリパターンにマッチするか否かを調べる

Endpoint 汎用

1 ASK {
2 <http://example.com/foo> ?p ?o .
3 }

ASKを使うと、クエリパターンにマッチする結果の有無をブール値の true/falseで得ることができ

ます。

第 2章 初級

4本目 DESCRIBEで指定した URIに関するデータを取得する

Endpoint https://ja.dbpedia.org/sparql

1 DESCRIBE ?s
2 WHERE {
3 ?s <http://www.w3.org/2000/01/rdf-schema#label> "ペタンク"@ja .
4 }

DESCRIBEは、指定した IRIのリソースに関する情報を含んだ RDFグラフを取得するクエリです。

サンプルクエリでは、WHEREで指定したパターンに従って、rdfs:labelの目的語に"ペタンク"@jaを

持つ主語のリソースに関する情報が得られます。

結果の内容は仕様上定義されておらず、SPARQLサーバの実装に依存します。例えば Virtuosoの場

合、デフォルトの設定では、指定した IRIを主語または目的語に持つトリプルが結果として返却されま

す。sql:describe-mode pragma*1を用いて、いくつかの異なる形式で結果を得ることができます。

5本目 FILTERで指定した範囲の値にマッチするリソースを取得する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT *
2 WHERE {
3 ?s <http://ja.dbpedia.org/property/生年> ?date .
4 FILTER (?date > "1950-01-01"^^xsd:date && ?date < "1959-12-31"^^xsd:date)
5 }
6 ORDER BY (?date)

FILTERとその条件に比較演算子「>」「<」を使って、1950年 1月 1日～1959年 12月 31日の間に

生まれた人物を取得しています。"1959-12-31"^^xsd:date は型付きリテラルを表しており、キャ

スト関数を使って xsd:date("1959-12-31")と書くこともできます。

また、論理和の演算子「&&」は Virtuosoでは「and」も使うことができますが、標準ではないので

注意してください。

6本目 リテラルの言語タグを取得する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT DISTINCT (LANG(?label) AS ?langtag)
2 WHERE {
3 ?s <http://www.w3.org/2000/01/rdf-schema#label> ?label .
4 }

LANG関数を使ってリテラルの言語タグを取得しています。言語タグがない場合は""が返されます。

*1 http://docs.openlinksw.com/virtuoso/rdfsqlfromsparqldescribe/

7本目 IFで条件分岐する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT DISTINCT ?name
2 (IF(isIRI(?kamon), STRAFTER(STR(?kamon), "resource/"), STR(?kamon)) AS ?kamonStr)
3 WHERE {
4 ?s <http://ja.dbpedia.org/property/家紋> ?page ;
5 <http://www.w3.org/2000/01/rdf-schema#label> ?name ;
6 <http://ja.dbpedia.org/property/家紋名称> ?kamon .
7 }

ある値で条件分岐したいときは、IF関数を使います。IF関数は、第一引数をブール値*2として評価

し、結果が trueであれば第二引数を返し、違えば第三引数を返します。

このクエリでは家紋の名称を取得したいのですが、<http://ja.dbpedia.org/property/家紋名称>の

目的語 (?kamon)は、IRIとリテラルが混在しています。isIRI関数で?kamonが IRIであるかを判定

し、IRIであれば STRAFTER関数で"resource/"以降の文字列を切り出します (STRAFTER関数は

文字列リテラルを引数とするので、?kamonを STR関数で変換していることに注意)。IRIでなければ

STR(?kamon)を返します。

8本目 BINDで値を変数にバインドする

Endpoint https://ja.dbpedia.org/sparql

1 SELECT DISTINCT ?name ?kamonStr
2 WHERE {
3 ?s <http://ja.dbpedia.org/property/家紋> ?page ;
4 <http://www.w3.org/2000/01/rdf-schema#label> ?name ;
5 <http://ja.dbpedia.org/property/家紋名称> ?kamon .
6 BIND (IF(isIRI(?kamon), STRAFTER(STR(?kamon), "resource/"), STR(?kamon))
7 AS ?kamonStr)
8 }

7本目と同じ結果が得られますが、IFを SELECT句に置くのではなく、WHERE句の中で BIND

を使って書くこともできます。複雑な条件分岐を書く場合は SELECT 句に置くと読みにくくなるの

で、BINDを使って整理して書くと良いでしょう。

9本目 REPLACEで文字列を正規表現で置換する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT DISTINCT ?name ?kamonStr
2 WHERE {
3 ?s <http://ja.dbpedia.org/property/家紋> ?page ;
4 <http://www.w3.org/2000/01/rdf-schema#label> ?name ;
5 <http://ja.dbpedia.org/property/家紋名称> ?kamon .

*2 17.2.2 Effective Boolean Value (EBV), https://www.w3.org/TR/sparql11-query/#ebv

6 BIND (IF(isIRI(?kamon), REPLACE(STR(?kamon), ".+/resource/([^/]+)$", "$1"), STR(?
kamon)) AS ?kamonStr)

7 }

このクエリも 7本目、8本目と同じ結果が得られます。STRAFTER関数を使って IRIから家紋名

称を取り出していましたが、このクエリでは REPLACE関数を使って同等の処理を実現しています。

REPLACE関数では第一引数の文字列を、第二引数を正規表現パターン、第三引数を置換文字列とし

て置換します。

正規表現をパターンをクエリ内に記述するときは、エスケープが必要な文字に注意してください。特

にバックスラッシュ (\)には注意が必要です。文字「\」にマッチさせたい場合は、まず SPARQLクエ

リ内の文字列に対するエスケープシーケンスで「\\」となり、さらに正規表現に対するエスケープシー

ケンスで「\\\\」とする必要があります。

10本目 REGEXで文字列を正規表現でマッチする

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?s ?label
2 WHERE {
3 ?s <http://www.w3.org/2000/01/rdf-schema#label> ?label .
4 FILTER REGEX(?label, "QL$", "i")
5 }

FILTERと REGEX関数を使って、指定した正規表現にマッチする?labelを持つリソースを取得し

ます。REGEX 関数の第三引数には"i"フラグが指定されていますので、大文字小文字問わず末尾が

「QL」である?labelを持つリソースが抽出されます。

グラフパターンではなく、文字列リテラルを評価した条件で FILTERするときに多用する関数なの

で、よく覚えておきましょう。

11本目 DBpediaのカテゴリのリストを取得する

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
2 SELECT ?s
3 WHERE {
4 ?s a skos:Concept.
5 FILTER CONTAINS(STR(?s), "resource/Category:")
6 }

skos:Conceptのインスタンスであるという条件のみだとカテゴリ以外のリソースも混じってしま

うので、CONTAINS 関数を使ってカテゴリを示す IRI かどうか判定します。CONTAINS 関数は、

REGEX関数のような正規表現パターンではなく単純に部分文字列として含まれるかどうかを判定し

ます。この場合は REGEX関数でも代替可能です。

12本目 カテゴリと同名の記事を取得する

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
3 SELECT ?article ?category
4 WHERE {
5 ?article rdfs:label ?labelArticle .
6 ?category a skos:Concept ;
7 rdfs:label ?labelConcept .
8 FILTER (CONTAINS(STR(?category), "resource/Category:"))
9 FILTER (?article != ?category)

10 FILTER (?labelArticle = ?labelConcept)
11 }
12 ORDER BY ?labelArticle

カテゴリを取得する方法は 11本目と同様です。記事とカテゴリのラベルをそれぞれ取得し、それら

が同じであるという条件を 10行目の FILTERで指定しています。2つが同じリソースである場合を排

除するために、9行目の FILTERが必要です。

13本目 都道府県のリストを取得する

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX dcterms: <http://purl.org/dc/terms/>
2 SELECT ?pref ?code
3 WHERE {
4 ?pref dcterms:subject <http://ja.dbpedia.org/resource/Category:日本の都道府県> ;
5 <http://ja.dbpedia.org/property/区分> ?category ;
6 <http://dbpedia.org/ontology/areaCode> ?code .
7 FILTER REGEX(?category, "[都道府県]")
8 FILTER REGEX(?code, "^JP")
9 }

10 ORDER BY (xsd:integer(STRAFTER(?code, "JP-")))

DBpedia では記事のカテゴリを dcterms:subject の目的語として取得できます。このクエリで

は、「日本の都道府県」カテゴリに属し prop-ja:区分が「都道府県」のいずれかである、というグラ

フパターンを指定することで、都道府県を取得しています。さらに、全国地方公共団体コードを?code

に取り出してソートしています。

14本目 バレーボールを題材としたアニメ作品を取得する

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX dcterms: <http://purl.org/dc/terms/>
2 PREFIX category-ja: <http://ja.dbpedia.org/resource/Category:>
3 SELECT DISTINCT ?s WHERE {
4 ?s dcterms:subject ?volley ;
5 dcterms:subject ?cat .
6 VALUES ?volley {category-ja:バレーボールを題材とした作品 category-ja:女子バレーボール

を題材とした作品 category-ja:バレーボール漫画}
7 FILTER CONTAINS(STR(?cat), "アニメ作品")
8 }

このクエリも DBpediaのカテゴリをうまく利用しています。6行目で、バレーボールを題材とした

作品に付与されるであろうカテゴリを列挙して、?volleyにバインドします。VALUESは BINDとは

異なり、複数の値を変数にバインドすることができます。4行目のパターンを評価すると、「バレーボー

ルを題材とした作品」「女子バレーボールを題材とした作品」「バレーボール漫画」のいずれかのカテゴ

リに属する、という条件になります。このクエリでは、VALUESを FILTERと INを用いて以下のよ

うに書き換えても同じ結果が得られます。

6 FILTER (?volley IN (category-ja:バレーボールを題材とした作品, category-ja:女子バレーボ
ールを題材とした作品, category-ja:バレーボール漫画))

7行目の FILTERは、「アニメ作品」という文字列を含むカテゴリでフィルタすることでアニメ作品

だけに絞り込んでいます。

15本目 プログラミング言語の一覧を取得する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT *
2 WHERE {
3 ?s a <http://dbpedia.org/ontology/ProgrammingLanguage> ;
4 <http://www.w3.org/2000/01/rdf-schema#label> ?label .
5 }
6 ORDER BY ?label

dbpedia-owl:ProgrammingLanguageのインスタンスであるリソースを取得しています。

16本目 プログラミング言語の一覧を頭文字で集計して降順に表示する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?initial (COUNT(?s) AS ?count) (GROUP_CONCAT(?label; SEPARATOR=", ") AS ?list)
2 WHERE {
3 ?s a <http://dbpedia.org/ontology/ProgrammingLanguage> ;
4 <http://www.w3.org/2000/01/rdf-schema#label> ?label .
5 BIND (SUBSTR(?label, 1, 1) AS ?initial)

6 }
7 GROUP BY ?initial
8 ORDER BY DESC(COUNT(?initial))

15本目に加えて、頭文字毎に集計しています。SUBSTRで先頭 1文字を切り出し、GROUP BYで

それを集約します。GROUP_CONCATは、集約された値を SEPARATORで連結した文字列を返し

ます。このクエリでは、頭文字が同じプログラミング言語のラベルが「, 」で連結されて表示されます。

17本目 特定のプロパティを N個以上持つリソースを取得する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?s (COUNT(?class) AS ?count)
2 WHERE {
3 ?s a ?class .
4 }
5 GROUP BY ?s
6 HAVING (COUNT(?class) > 14)
7 ORDER BY DESC(COUNT(?class))

GROUP BYで集約を作成し、集約された結果に対する条件は FILTERではなく HAVINGを使っ

て指定します。このクエリでは、クラスを 15個以上持つリソースを取得しています。

18本目 今日が誕生日の人を取得する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT *
2 WHERE {
3 ?s <http://ja.dbpedia.org/property/生月> ?month;
4 <http://ja.dbpedia.org/property/生日> ?day .
5 FILTER (?month = MONTH(NOW()) && ?day = DAY(NOW()))
6 }

NOWやMONTHなどの日時に関する関数を使って、今日が誕生日の人物を取得しています。比較

を行うときは型に注意しましょう。?month、?dayは整数 (xsd:integer)の値を持ち、MONTH関

数と DAY関数も整数が返り値です。

19本目 東京都出身の 50歳以下の俳優を取得する

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>
2 PREFIX dcterms: <http://purl.org/dc/terms/>
3 PREFIX category-ja: <http://ja.dbpedia.org/resource/Category:>
4 PREFIX prop-ja: <http://ja.dbpedia.org/property/>
5 PREFIX dbpedia-ja: <http://ja.dbpedia.org/resource/>
6 SELECT * WHERE {
7 ?actor prop-ja:職業 dbpedia-ja:俳優 ;

8 dcterms:subject category-ja:東京都出身の人物 ;
9 prop-ja:生年 ?year.

10 OPTIONAL {?actor prop-ja:出生地 ?birthplace}
11 FILTER (?year >= (YEAR(NOW()) - 50))
12 FILTER (DATATYPE(?year) = xsd:integer)
13 }
14 ORDER BY ?year

「東京都出身の人物」カテゴリと prop-ja:職業を使うことで、東京出身の俳優を取得できます。

prop-ja:生年で得られる生年をもとに 50歳以下という条件で絞り込みます。また、?yearが文字列で

「不詳」となっているデータもあるため、12行目で型が xsd:integerでないものを除外しています。

20本目 職業に「声優」を含む人名典拠を取得する

Endpoint http://id.ndl.go.jp/auth/ndla/sparql

1 PREFIX rda: <http://RDVocab.info/ElementsGr2/>
2 PREFIX foaf: <http://xmlns.com/foaf/0.1/>
3

4 SELECT *
5 WHERE {
6 ?concept foaf:primaryTopic ?topic .
7 ?topic a foaf:Person ;
8 foaf:name ?name ;
9 rda:biographicalInformation ?work .

10 FILTER CONTAINS(?work, "声優")
11 }

エンドポイントは DBpedia Japanese ではなくWeb NDL Authorities(NDLA) です。クエリ自体

は簡単ですが、NDLAの典拠データのモデルを理解していないと書きづらいため紹介します。

NDLA の個人名・家族名・団体名の典拠データは、典拠情報とその記述対象である名称実体で構

成されます*3。6 行目の?concept が典拠情報にあたり、7 行目の?topic が名称実体にあたります。

rda:biographicalInformation から職業を取得し、「声優」を含むもののみを抽出しています (9,

10行目)。

21本目 プロパティパス InversePathを使う

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?s
2 WHERE {
3 ?s ^<http://dbpedia.org/ontology/wikiPageWikiLink> <http://ja.dbpedia.org/resource/

日本の山一覧> .
4 }

プロパティパス式「^」は逆向き (目的語から主語)の向きに述語を辿ります。つまり、以下のグラフ

パターンと等価です。

*3 RDFモデルについて, http://id.ndl.go.jp/information/model/

<http://ja.dbpedia.org/resource/日本の山一覧> <http://dbpedia.org/ontology/
wikiPageWikiLink> ?s .

dbpedia-owl:wikiPageWikiLinkは、Wikipedia記事中の他の記事へのリンクを示します。従っ

て、「日本の山一覧」からリンクを張られている記事のリソース IRIが?sとして得られます。

22本目 乱数を生成する

Endpoint 汎用

1 SELECT (RAND() AS ?rand)
2 WHERE {
3 ?s ?p ?o .
4 }
5 LIMIT 10

RAND関数を実行すると [0, 1)の一様乱数が生成されます。しかし、乱数生成の挙動は実装に依存
することが多いので注意が必要です。

例えば、DBpedia で使われている Virtuoso でこのクエリを実行すると、同じ数値が 10 個出力さ

れてしまいます。原因の解説については初級の範囲を超えてしまいますので、後述することにします

(�48本目)。これを回避するためには RAND関数ではなく、Virtuosoの独自関数 bif:rndを使う方法

があります (�38本目)。

第 3 章

中級より複雑なクエリで理解を深めよう！

初級のノックを踏まえて、GROUP BYによる集約やプロパティパスを使った複雑なグラフパター

ンに挑戦します。

23本目 DBpediaから上位カテゴリを持たないカテゴリを取得する

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
2 SELECT distinct ?s WHERE {
3 ?s a skos:Concept .
4 FILTER NOT EXISTS {?s skos:broader ?bc}
5 FILTER REGEX(STR(?s), ’dbpedia.org/resource/Category’)
6 }

11本目をベースに、上位カテゴリを持つカテゴリのパターンを NOT EXISTSをに指定して結果か

ら除外します。

24本目 クラスの IRIのローカル名っぽい部分をリストにする

Endpoint 汎用

1 SELECT DISTINCT
2 ?prefix (COUNT(DISTINCT(?local)) AS ?num)
3 (GROUP_CONCAT(DISTINCT ?local; SEPARATOR=", ") AS ?localNames)
4 WHERE {
5 ?s a ?o .
6 BIND (STR(?o) AS ?oStr)
7 BIND (REPLACE(?oStr, ".+[^a-zA-Z_0-9-]([A-Za-z0-9_-]*)$", "$1") AS ?local)
8 BIND (STRBEFORE(?oStr, ?local) AS ?prefix)
9 }

10 GROUP BY ?prefix
11 ORDER BY DESC(?num)

1本目ではすべてのクラス IRIを取得しましたが、このクエリでは IRIからローカル名*1を分離し、

接頭辞毎に集計します。また、GROUP BYと GROUP_CONCATを使って集約されたローカル名

を表示します。

7行目の REPLACEで正規表現を使ってローカル名を抽出していますが、この正規表現は SPARQL

仕様のローカル名と厳密には対応していません。詳細は定義*2を参照してください。

*1 実際には接頭辞宣言によるため、あくまでそれっぽい部分です
*2 https://www.w3.org/TR/sparql11-query/#rPN_LOCAL

25本目 特定のプロパティを N個以上持つリソースを取得する (サブクエリ版)

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?s ?count
2 WHERE {
3 FILTER (?count > 14)
4 {
5 SELECT ?s (COUNT(?type) AS ?count)
6 WHERE {
7 ?s a ?type .
8 }
9 GROUP BY ?s

10 }
11 }
12 ORDER BY DESC(?count)

17 本目と同じことをサブクエリを使って書いています。このクエリのように単純な条件であれば、

HAVINGを使ったほうが短く簡潔で良いでしょう。

26本目 夏季冬季オリンピック金メダリスト出身地別リストを取得する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?pref (count(?name) AS ?count) (GROUP_CONCAT(?name; separator=’,’) AS ?people)
2 WHERE {
3 ?person <http://purl.org/dc/terms/subject> <http://ja.dbpedia.org/resource/Category:

日本のオリンピック金メダリスト> ;
4 <http://purl.org/dc/terms/subject> ?bplace ;
5 <http://www.w3.org/2000/01/rdf-schema#label> ?name .
6 ?bplace <http://www.w3.org/2000/01/rdf-schema#label> ?pref .
7 FILTER REGEX(?pref, "[都道府県州国]出身の人物", "i")
8 }
9 GROUP BY (?pref)

10 ORDER BY DESC (count(?person))

こちらも GROUP BYと GROUP_CONCATを活用した集計です。

27本目 日本人の姓（っぽい箇所）を取り出して集計する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?family (COUNT(?family) AS ?count)
2 WHERE {
3 ?s a <http://dbpedia.org/ontology/Person>;
4 <http://dbpedia.org/ontology/nationality> <http://ja.dbpedia.org/resource/日本> ;
5 <http://www.w3.org/2000/01/rdf-schema#comment> ?comment .
6 BIND (REPLACE(?comment, "([ ・　(（]|は、|は日本).*$", "") AS ?family)
7 FILTER (STRLEN(?family) > 0)
8 }
9 GROUP BY ?family

10 ORDER BY DESC(COUNT(?family))

DBpediaで姓のみを取得できるプロパティがあればよいのですが、元のWikipedia記事の記述から

うまく切り出すしかありません。記事冒頭の書き出しには一定のパターンがあるので、正規表現を使っ

て姓と思われる箇所を取り出しています (6行目)。姓名が一緒に取り出されてしまったり多少のゴミが

含まれていたりするので、もう少し工夫の余地はありそうです。

28本目 世界遺産のリストを取得する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?heritage ?name ?country ?lat ?long
2 WHERE {
3 ?heritage <http://purl.org/dc/terms/subject> ?category ;
4 <http://www.w3.org/2000/01/rdf-schema#label> ?name ;
5 <http://ja.dbpedia.org/property/country> ?country .
6 ?category <http://www.w3.org/2004/02/skos/core#broader> <http://ja.dbpedia.org/

resource/Category:五十音順の世界遺産> .
7 OPTIONAL {
8 ?heritage <http://www.w3.org/2003/01/geo/wgs84_pos#lat> ?lat ;
9 <http://www.w3.org/2003/01/geo/wgs84_pos#long> ?long .

10 }
11 }
12 ORDER BY (?country)

カテゴリ「五十音順の世界遺産」は下位カテゴリに「世界遺産 あ行」～「世界遺産 わ行」という下

位カテゴリを持ち、そのカテゴリから個別の世界遺産のページを辿ることができます。従って、3, 6行

目で「五十音順の世界遺産」が上位カテゴリであるカテゴリを持つリソースを取得するパターンを指定

しています。また、緯度・経度を持たないリソースも存在するため、OPTIONAL内のパターンとして

指定しておきます。

29本目 借用語のリストを取得する

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX dcterms: <http://purl.org/dc/terms/>
2 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
3 PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
4

5 SELECT ?category (COUNT(?word) AS ?count)
6 (GROUP_CONCAT(?word; SEPARATOR=’, ’) AS ?wordList)
7 WHERE {
8 ?s dcterms:subject ?category ;
9 rdfs:label ?word .

10 ?category skos:broader <http://ja.dbpedia.org/resource/Category:借用語> ;
11 rdfs:label ?country .
12 FILTER NOT EXISTS {
13 ?s dcterms:subject <http://ja.dbpedia.org/resource/Category:日本語における借用語> .
14 }

– 19 –

https://ja.dbpedia.org/sparql
https://ja.dbpedia.org/sparql


15 }
16 GROUP BY (?category)
17 ORDER BY DESC(COUNT(?s))

カテゴリ「借用語」の下位カテゴリから借用語を取得し、カテゴリ毎にGROUP BYで集計していま

す。「日本語における借用語」カテゴリのリソースは借用語そのものを扱ったものではないので、NOT

EXISTSのパターンで除外しています (13行目)。

30本目 政党のイデオロギーと政治的思想・立場を取得する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?s ?name (COUNT(DISTINCT(?ideology)) AS ?count)
2 (GROUP_CONCAT(DISTINCT ?ideology; separator=", ") AS ?ideologies)
3 WHERE {
4 ?s <http://dbpedia.org/ontology/ideology>|<http://ja.dbpedia.org/property/政治的思想

・立場> ?i ;
5 <http://www.w3.org/2000/01/rdf-schema#label> ?name .
6 ?i <http://www.w3.org/2000/01/rdf-schema#label> ?ideology .
7 }
8 ORDER BY DESC(COUNT(DISTINCT(?ideology)))

プロパティパスを使って、述語が dbpedia-owl:ideologyまたは prop-ja:政治的思想・立場の目

的語を取得しています。両者の目的語は重複した値を持ち得るので、COUNTやGROUP_CONCAT

では DISTINCTを忘れないようにします。

31本目 複数の都道府県にまたがる日本の山のリストを取得する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?s (COUNT(DISTINCT(?pref)) AS ?count)
2 (GROUP_CONCAT(DISTINCT ?pref; SEPARATOR=’, ’) AS ?prefList)
3 WHERE {
4 ?s a <http://dbpedia.org/ontology/Mountain> ;
5 <http://dbpedia.org/ontology/address> ?address ;
6 <http://www.w3.org/2000/01/rdf-schema#label> ?name ;
7 ^<http://dbpedia.org/ontology/wikiPageWikiLink> <http://ja.dbpedia.org/resource/日

本の山一覧> .
8 BIND (REPLACE(STR(REPLACE(?address, "([（）]|藤津郡太良町・)", "")), "(京都府|[都道府

県]).*$", "$1") AS ?pref)
9 FILTER (REGEX(?pref, "[都道府県]"))

10 FILTER (!REGEX(?pref, "(同県|小県)"))
11 }
12 GROUP BY ?s
13 HAVING (COUNT(DISTINCT(?pref)) > 1)
14 ORDER BY DESC (COUNT(DISTINCT(?pref)))

プロパティパス式「^dbpedia-owl:wikiPageWikiLink」を使って、「日本の山一覧」からリンク

されている山のリストを抽出しています。山の所在地の都道府県名のみを取得できるプロパティが存

在しないため、8～10行目では、?addressから正規表現を使って都道府県名を取り出しています。

32本目 コレクション (rdf:List)からメンバ (rdf:first)を取り出す

Endpoint 汎用

1 PREFIX ex: <http://example.org/>
2 PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
3

4 SELECT *
5 WHERE {
6 ?s ex:member/rdf:rest*/rdf:first ?o .
7 }

RDFコレクションはいわゆる線形リストのデータ構造で、リンクを辿って順にノードを参照します。

SAPRQLではプロパティパスを使うことで、簡単に各メンバを取り出すことができます。「/」は指定

の述語を順にたどり、「*」は述語の 0回以上繰り返しを意味します。

例えば、以下のようなデータが与えられたときを考えましょう。

1 @prefix ex: <http://example.org/> .
2 @prefix rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#> .
3

4 <http://example.org/group/1> ex:member _:bn1 .
5 _:bn1 rdf:first <http://example.org/member/foo> .
6 _:bn1 rdf:rest _:bn2 .
7 _:bn2 rdf:first <http://example.org/member/bar> .
8 _:bn2 rdf:rest _:bn3 .
9 _:bn3 rdf:first <http://example.org/member/baz> .

10 _:bn3 rdf:rest rdf:nil .

サンプルクエリをこのデータに対して実行すると、?oには<http://example.org/member/foo>、

<http://example.org/member/bar>、　<http://example.org/member/baz>が得られます。た

だし、得られる結果の順序は保証されないことに注意してください。

33本目 コレクションから任意の位置のメンバを取り出す

Endpoint 汎用

1 PREFIX ex: <http://example.org/>
2 PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
3

4 SELECT *
5 WHERE {
6 ?s ex:member/rdf:rest{2}/rdf:first ?o .
7 }

プロパティパス式「rdf:rest{n}」で繰り返し回数を指定することで、任意の位置のメンバを取り出す

ことができます。このクエリの場合は rdf:restを 2回辿ることになりますので、32本目と同じデータ

を与えたとすると、?oは<http://example.org/member/baz>になります。

ただし、このプロパティパス式は現在のW3C 勧告の仕様には存在しません。ドラフトの仕様*3で

*3 SPARQL 1.1 Property Paths - W3C Working Draft 26 January 2010,

は存在していたようで、Virtuosoや Apache Jena*4など主要な実装では拡張としてサポートされてい

ます。

34本目 あるカテゴリの下位カテゴリに属する記事を取得する

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
2 PREFIX dcterms: <http://purl.org/dc/terms/>
3

4 SELECT DISTINCT ?s ?category
5 WHERE {
6 ?s dcterms:subject ?category .
7 <http://ja.dbpedia.org/resource/Category:アニメ> ^skos:broader+ ?category .
8 }

プロパティパス式「^skos:broader+」は、skos:broaderの 1回以上の繰り返しのパスを目的語

から主語の方向に辿るパスを意味します。7行目で「アニメ」カテゴリの下位カテゴリを取得し、6行

目でそのカテゴリを持つ記事を取得します。

このように一つの述語に複数の演算子を適用することができますが、優先順位に注意してください

(この例ではどちらを先に評価しても結果は同じです)。

35本目 複数のグラフをマージしたグラフに問い合わせる

Endpoint 汎用

1 SELECT ?s ?o
2 FROM <http://example.org/foo>
3 FROM <http://example.org/bar>
4 FROM <http://example.org/baz>
5 WHERE {
6 ?s <http://example.org/term> ?o .
7 }

FROM句はデフォルトグラフとするグラフを示す IRIを指定します。複数の FROM句を指定した

場合は、それらがマージされたグラフがデフォルトグラフとなります。このクエリの例では 3つの IRI

が FROM句で指定されており、これらのグラフがマージされたグラフを対象にクエリが実行されるこ

とになります。

一方で、FROM句を指定しない場合のクエリ対象は、SPARQL サーバによって異なる場合がある

ので注意してください。Virtuosoではすべての名前付きグラフがマージされたグラフが対象となりま

す。*5。Fusekiではデータセットで定義されているデフォルトグラフが対象となります*6(TDBの設定

tdb:unionDefaultGraph*7ですべての名前付きグラフをデフォルトグラフとするように変更可能)。

https://www.w3.org/TR/2010/WD-sparql11-property-paths-20100126/#path-language
*4 ARQ - Property Paths, https://jena.apache.org/documentation/query/property_paths.html
*5 https://github.com/openlink/virtuoso-opensource/issues/616#issuecomment-268542782
*6 TDB Datasets, http://jena.apache.org/documentation/tdb/datasets.html
*7 TDB Configuration, https://jena.apache.org/documentation/tdb/configuration.html

さらに Fusekiでは、デフォルトグラフを示す<urn:x-arq:DefaultGraph>と全ての名前付きグラ

フがマージされたグラフを示す<urn:x-arq:UnionGraph>という特別なグラフ IRI が与えられてい

ます。次のクエリのように FROM句で指定すれば、データセット内の全てのトリプルに問い合わせす

ることが可能です。

1 SELECT *
2 FROM <urn:x-arq:DefaultGraph>
3 FROM <urn:x-arq:UnionGraph>
4 WHERE {
5 ?s <http://example.org/term> ?o .
6 }

名前付きグラフがアプリケーションなどで動的に生成されるケースなどで役に立つかもしれません。

36本目 ジャニーズ事務所所属タレントと 4回以上共演した人を取得する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT DISTINCT ?p1 ?p2
2 WHERE {
3 ?work1 <http://dbpedia.org/ontology/starring> ?p1 ;
4 <http://dbpedia.org/ontology/starring> ?p2 .
5 ?work2 <http://dbpedia.org/ontology/starring> ?p1 ;
6 <http://dbpedia.org/ontology/starring> ?p2 .
7 ?work3 <http://dbpedia.org/ontology/starring> ?p1 ;
8 <http://dbpedia.org/ontology/starring> ?p2 .
9 ?work4 <http://dbpedia.org/ontology/starring> ?p1 ;

10 <http://dbpedia.org/ontology/starring> ?p2 .
11 ?p1 <http://ja.dbpedia.org/property/production> <http://ja.dbpedia.org/resource/ジャ

ニーズ事務所> .
12 FILTER (?work1 != ?work2)
13 FILTER (?work1 != ?work3)
14 FILTER (?work1 != ?work4)
15 FILTER (?work2 != ?work3)
16 FILTER (?work2 != ?work4)
17 FILTER (?work3 != ?work4)
18 FILTER (?p1 != ?p2)
19 }

少し長めなので順に説明します。

• 3～10 行目 出演を表す述語 dbpedia-owl:starring を使って、共演関係にある 2 人の人物

(?p1, ?p2) を抽出する。出演作品?work1 から?work4 を 4 つ並べて、4 回以上共演している人

物とする。

• 11行目 片方の人物 (?p1)はジャニーズ事務所所属である

• 12～17行目 4つの出演作品は重複がないという条件

• 18行目 ?p1と?p2は同一人物ではない

　このクエリを実行すると、執筆時点では、18組の結果が得られました。うまく結果が得られたと思

われるかもしれませんが、一つ落とし穴があるのです（人物だけでなく TOKIOや嵐などのグループ

が含まれているという問題もありますが、ここでは目を瞑ります）。

結果をよく見ると、次のような重複する組が存在しているのです。

p1 p2

http://ja.dbpedia.org/resource/城島茂 http://ja.dbpedia.org/resource/松岡昌宏

http://ja.dbpedia.org/resource/松岡昌宏 http://ja.dbpedia.org/resource/城島茂

SPARQL ではグラフパターンにマッチするデータが全て返されるので、このクエリの条件のみで

は?p1 と?p2 の値が逆になった結果が含まれます。なお、SELECT DISTINCT で重複排除されるの

は、あくまで結果の組に対してです。

FILTER NOT EXISTS {
?p2 <http://ja.dbpedia.org/property/production> <http://ja.dbpedia.org/resource/ジャ
ニーズ事務所>

}

という条件を追加すればよいと思われるかもしれませんが、これではジャニーズ事務所所属タレント

同士が共演する場合も取り除かれてしまいます。つまり、「ジャニーズ事務所所属タレントと 4回以上

共演したジャニーズ事務所所属タレントでない人」というお題になってしまいます。

これを排除するにはかなり工夫が必要で、詳しくは 37本目をご覧ください。

37本目 ジャニーズ事務所所属タレントと 4回以上共演した人を取得する (重複排除)

Endpoint https://ja.dbpedia.org/sparql

1 SELECT DISTINCT ?pp1 ?pp2
2 WHERE {
3 ?work1 <http://dbpedia.org/ontology/starring> ?p1 ;
4 <http://dbpedia.org/ontology/starring> ?p2 .
5 ?work2 <http://dbpedia.org/ontology/starring> ?p1 ;
6 <http://dbpedia.org/ontology/starring> ?p2 .
7 ?work3 <http://dbpedia.org/ontology/starring> ?p1 ;
8 <http://dbpedia.org/ontology/starring> ?p2 .
9 ?work4 <http://dbpedia.org/ontology/starring> ?p1 ;

10 <http://dbpedia.org/ontology/starring> ?p2 .
11 ?p1 <http://ja.dbpedia.org/property/production> <http://ja.dbpedia.org/resource/ジャ

ニーズ事務所> .
12 FILTER (?work1 NOT IN (?work2, ?work3, ?work4))
13 FILTER (?work2 NOT IN (?work3, ?work4))
14 FILTER (?work3 != ?work4)
15 FILTER (?p1 != ?p2)
16 BIND (IF(STR(?p1) <= STR(?p2), ?p1, ?p2) AS ?pp1)
17 BIND (IF(STR(?p1) > STR(?p2), ?p1, ?p2) AS ?pp2)
18 }

36 本目の結果に存在していた重複する組を排除する条件を追加したクエリです。16,17 行目の条件

がポイントです。この IF で?p1, ?p2 の順序を比較することで、同じ値は必ず同じ変数にバインドさ

れるということを保証しています。{?p1 := foo, ?p2 := bar}あるいは{?p1 := bar, ?p2 :=

foo}が与えられたとき、必ず{?pp1 := foo, ?pp2 := bar}になるということです。このようにす

れば、SELECT DISTINCTで重複が排除されます。

本題ではありませんが、?work1,　?work2,　?work3,　?work4間で重複がないという条件をより

短く書くことにも挑戦してみたのですが、12～14行目のように NOT INを使う以上にうまい方法は思

い付きませんでした。これでは 36本目のように「!=」で全通り羅列しているのと大差ないですね。

第 4 章

上級様々な構文や関数を使って自在にクエリを組み立てよう！

これまでのノックを踏まえて応用的なテクニックを紹介します。対象とするデータセットの構造に

合わせて、柔軟にクエリを組み立てる能力を養います。

38本目 乱数を生成する (Virtuoso用)

Endpoint 汎用

1 SELECT (bif:rnd(10, ?s, ?p, ?o) AS ?rand)
2 WHERE {
3 ?s ?p ?o .
4 }
5 LIMIT 10

Virtuosoの独自関数 bif:rnd()*1を使って乱数を生成します。22本目で扱った RAND関数とは仕様

が異なり、第一引数に指定した整数未満 0以上の範囲で乱数を生成します。第一引数を指定しただけで

は、22本目と同様に同じ数値が出力されてしまいます。第二引数以降に?s, ?p, ?oを渡すことで、毎回

異なる乱数が生成されるようになります。

どうしても RAND関数を使いたい場合は、以下のように少しダーティなハックが必要です*2。

1 SELECT (RAND(1 + strlen(str(?s))*0) AS ?rand)
2 WHERE {
3 ?s ?p ?o .
4 }
5 LIMIT 10

ただし、標準の RAND 関数は引数を取らないためこのクエリでも仕様に違反します。Virtuoso の

実装では、RAND関数の第一引数は double型の数値をとり、その数値未満の乱数を生成するようで

す。Virtuosoは文書化されていない仕様が多くて困りますね…。

39本目 SPARQLどうでしょう「サイコロの旅」

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?line ?stations
2 WHERE {
3 <http://ja.dbpedia.org/resource/秋葉原駅> <http://dbpedia.org/ontology/

servingRailwayLine> ?line .

*1 http://docs.openlinksw.com/virtuoso/rndsalltr/
*2 How to select random DBPedia nodes from SPARQL?,

https://stackoverflow.com/questions/5677340/how-to-select-random-dbpedia-nodes-from-sparql

4 ?stations <http://dbpedia.org/ontology/servingRailwayLine> ?line .
5 }
6 ORDER BY (RAND(1 + STRLEN(?stations)*0))
7 LIMIT 1

このクエリでは乱数を生成する RAND関数を活用して、秋葉原駅から乗る路線とその路線での降車

駅をランダムに選択する「サイコロの旅」を SPARQLで実現します。38本目で紹介した方法で乱数

を生成しています。

40本目 パスの深さをカウントする

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
2 SELECT ?gchild (COUNT(?child)-1 AS ?depth)
3 WHERE {
4 <http://ja.dbpedia.org/resource/Category:アニメ> ^skos:broader* ?child .
5 ?child ^skos:broader* ?gchild .
6 }
7 GROUP BY ?gchild
8 ORDER BY ?depth

プロパティパスと GROUP BYをうまく使って述語をいくつ辿ったかをカウントすることができま

す。このクエリでは、カテゴリ「アニメ」以下のカテゴリとその深さを集計しています。下位カテゴリ

を取得する方法は 34本目と同じです。

より簡単なデータで考えてみましょう。以下のグラフは、:baz ¸:bar ¸:foo と skos:broader

でリンクしている構造です。

1 @prefix : <http://example.com/> .
2 @prefix skos: <http://www.w3.org/2004/02/skos/core#> .
3

4 :baz skos:broader :bar .
5 :bar skos:broader :foo .

このとき、<http://example.com/foo>を起点に skos:broaderの逆を辿ってみましょう。

1 PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
2 SELECT *
3 WHERE {
4 <http://example.com/foo> ^skos:broader* ?child .
5 ?child ^skos:broader* ?gchild .
6 }

?childと?gchildを出力してみると以下のようになります。「*」は 0以上の繰り返しですので、?child

と?gchild には foo から終点のノードまでのノードの組み合わせが出力されます。これを?gchild で

GROUP BYして COUNTすれば、fooからの深さとなります。

chilld gchild

<http://example.com/foo> <http://example.com/foo>

<http://example.com/foo> <http://example.com/bar>

<http://example.com/foo> <http://example.com/baz>

<http://example.com/bar> <http://example.com/bar>

<http://example.com/bar> <http://example.com/baz>

<http://example.com/baz> <http://example.com/baz>

41本目 31本目のクエリ結果を都道府県毎にリスト化する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?pref (COUNT(?mount) AS ?count)
2 (GROUP_CONCAT(?name; SEPARATOR=’, ’) AS ?mountList)
3 WHERE {
4 ?s a <http://dbpedia.org/ontology/AdministrativeRegion> ;
5 <http://www.w3.org/2000/01/rdf-schema#label> ?pref .
6 FILTER CONTAINS(?mountPref, STR(?pref))
7 {
8 SELECT ?mount ?name (GROUP_CONCAT(DISTINCT(?pref); SEPARATOR=’, ’) AS ?mountPref)
9 WHERE {

10 ?mount a <http://dbpedia.org/ontology/Mountain> ;
11 <http://dbpedia.org/ontology/address> ?address ;
12 <http://www.w3.org/2000/01/rdf-schema#label> ?name ;
13 ^<http://dbpedia.org/ontology/wikiPageWikiLink> <http://ja.dbpedia.org/

resource/日本の山一覧> .
14 BIND (REPLACE(STR(REPLACE(?address, "([（）]|藤津郡太良町・)", "")), "(京都府|[都

道府県]).*$", "$1") AS ?pref)
15 FILTER (REGEX(?pref, "[都道府県]"))
16 FILTER (!REGEX(?pref, "(同県|小県)"))
17 }
18 GROUP BY ?mount ?name
19 HAVING (COUNT(DISTINCT(?pref)) > 1)
20 }
21 }
22 GROUP BY ?pref
23 ORDER BY DESC(COUNT(?mount))

31本目では、山ごとに集約を作って都道府県のリストを作っていました。このクエリは都道府県毎

に集約し、その都道府県にまたがる山をリストにして出力します。

7行目からのサブクエリでは 31本目と同様に山毎の集約を作ります。外側のクエリは、

• 4,5行目 都道府県を取得するパターン

• 6 行目 サブクエリで得られた山の所在都道府県のリスト (?mountPref) 中に都道府県名が存在

するかを判定

• 18,19行目 都道府県名で集約

という構成になっていて、サブクエリと外側のクエリで二重に集約を作っているところがポイント

– 28 –

https://ja.dbpedia.org/sparql


第 4章 上級

です。

42本目 動的に IRIを生成する

Endpoint https://ja.dbpedia.org/sparql

1 SELECT DISTINCT ?nameProp ?name
2 WHERE {
3 ?location ?nameProp ?name .
4 {
5 SELECT DISTINCT (IRI(?prop) AS ?nameProp)
6 WHERE {
7 ?s <http://ja.dbpedia.org/property/subdivisionType> ?o .
8 BIND (CONCAT(REPLACE(STR(?o), "resource", "property"), "名"^^xsd:string) AS ?prop

)
9 }

10 }
11 }

動的に IRIを生成するというテクニックを紹介します。一般的にはどのような語彙があるか知った

上でクエリを書きますが、存在するか分からないが存在するものだけ取得したい場合には、このテク

ニックが活用できます。

このクエリは、DBpediaから地名を取得したいが地名を目的語とする述語がよく分からない、とい

うケースです。6～9行目のサブクエリで、地名を目的語に持ち得る述語の IRIを動的に生成していま

す (?nameProp)。3行目ではそれを述語としたパターンを記述しています。

43本目 スポーツを題材にしたアニメ作品を取得する

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX dcterms: <http://purl.org/dc/terms/>
2 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
3 PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>
4 PREFIX dbpedia-ja: <http://ja.dbpedia.org/resource/>
5

6 SELECT DISTINCT ?title ?category
7 WHERE {
8 ?s dcterms:subject ?category, ?categories ;
9 rdfs:label ?title .

10 FILTER CONTAINS(STR(?categories), "アニメ")
11 {
12 SELECT ?category
13 WHERE {
14 {
15 dbpedia-ja:スポーツ競技一覧 dbpedia-owl:wikiPageWikiLink/rdfs:label ?label .
16 BIND (IRI(CONCAT("http://ja.dbpedia.org/resource/Category:"^^xsd:string, ?label

, "アニメ"^^xsd:string)) AS ?category)
17 } UNION {
18 dbpedia-ja:スポーツ競技一覧 dbpedia-owl:wikiPageWikiLink/rdfs:label ?label .
19 BIND (IRI(CONCAT("http://ja.dbpedia.org/resource/Category:男子"^^xsd:string, ?

label, "アニメ"^^xsd:string)) AS ?category)
20 } UNION {
21 dbpedia-ja:スポーツ競技一覧 dbpedia-owl:wikiPageWikiLink/rdfs:label ?label .
22 BIND (IRI(CONCAT("http://ja.dbpedia.org/resource/Category:女子"^^xsd:string, ?

label, "アニメ"^^xsd:string)) AS ?category)
23 } UNION {
24 dbpedia-ja:スポーツ競技一覧 dbpedia-owl:wikiPageWikiLink/rdfs:label ?label .
25 BIND (IRI(CONCAT("http://ja.dbpedia.org/resource/Category:"^^xsd:string, ?label

, "を題材とした作品"^^xsd:string)) AS ?category)
26 } UNION {
27 dbpedia-ja:スポーツ競技一覧 dbpedia-owl:wikiPageWikiLink/rdfs:label ?label .
28 BIND (IRI(CONCAT("http://ja.dbpedia.org/resource/Category:男子"^^xsd:string, ?

label, "を題材とした作品"^^xsd:string)) AS ?category)
29 } UNION {
30 dbpedia-ja:スポーツ競技一覧 dbpedia-owl:wikiPageWikiLink/rdfs:label ?label .
31 BIND (IRI(CONCAT("http://ja.dbpedia.org/resource/Category:女子"^^xsd:string, ?

label, "を題材とした作品"^^xsd:string)) AS ?category)
32 }
33 }
34 }
35 }

長いクエリですが、同じようなグラフパターンを UNIONで繋げているだけなので、そこまで難し

いクエリではありません。12行目からのサブクエリのWHEREでやっていることは、42本目のテク

ニックを使って、スポーツを題材にしたアニメ作品に該当しそうなカテゴリの IRIを動的に生成して

います。以下のようなテンプレートで (6× スポーツ競技の名称 (?label) の数) 通りのカテゴリの IRI

が作成されます。

http://ja.dbpedia.org/resource/Category:(|男子|女子){?label}(アニメ|を題材とした作品)

ちなみに CONCAT関数の引数のリテラルをそれぞれ^^xsd:stringの型付きリテラルにしていますが、

通常のリテラルにすると正しく動きません。原因は調査中なのですが、Virtuosoの文字列操作関数の

実装に何らかの問題があると思われます。

そして、8～10行目では生成した IRIを使ってアニメ作品のリソースを絞り込んでいます。この点は

14本目と同じやり方です。

44本目 複数のパターンの直積によって IRIを生成する

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>
3

4 SELECT (IRI(
5 CONCAT("http://ja.dbpedia.org/resource/Category:", ?x, ?label, ?y)
6 ) AS ?category)
7 WHERE {
8 <http://ja.dbpedia.org/resource/スポーツ競技一覧> dbpedia-owl:wikiPageWikiLink/rdfs:

label ?label .
9 VALUES ?x {"" "男子" "女子"}

10 VALUES ?y {"アニメ" "を題材とした作品"}
11 }

43本目を改良して、長ったらしい UNIONを使わずに様々な IRIのパターンを生成する方法を考え

ます。43本目で 6通りのグラフパターンを UNIONしていたのは、スポーツ競技名称の先頭に「」「男

子」「女子」のいずれかが付き、末尾に「アニメ」「を題材とした作品」が付くカテゴリ名を生成するこ

とが目的でした。これらのパターンは{"", "男子", "女子"}と{"アニメ", "を題材とした作品"}と

いう 2つの集合の直積を考えれば、より短く書くことができます。

9行目の?xは生成するカテゴリ名のプレフィックスにあたり、10行目の yはカテゴリ名のサフィッ

クスにあたる文字列をバインドしています。直積を作り出すために 2 つの VALUES で定義してい

ることに注意してください。単に 2 つの変数の組に値をバインドする VALUES (?x ?y) {("foo"

"bar") ...} の構文とは異なります。そして 5行目で?x, ?label, ?yを結合してカテゴリの IRIを動

的に生成しています。

次のノックでは、この方法で生成したカテゴリの IRI を使ってアニメ作品を抽出しましょう (�45

本目)。

45本目 スポーツを題材にしたアニメ作品を取得する (44本目の方法を利用)

Endpoint https://ja.dbpedia.org/sparql

1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX dbpedia-owl: <http://dbpedia.org/ontology/>
3 SELECT DISTINCT ?title ?category
4 WHERE {
5 ?s <http://purl.org/dc/terms/subject> ?category, ?categories ;
6 rdfs:label ?title .
7 FILTER CONTAINS(STR(?categories), "アニメ")
8 {
9 SELECT (IRI(

10 CONCAT("http://ja.dbpedia.org/resource/Category:", ?x, ?label, ?y)
11 ) AS ?category)
12 WHERE {
13 <http://ja.dbpedia.org/resource/スポーツ競技一覧> dbpedia-owl:wikiPageWikiLink/

rdfs:label ?label .
14 VALUES ?x {"" "男子" "女子"}
15 VALUES ?y {"アニメ" "を題材とした作品"}
16 }
17 }
18 }

サブクエリの中身は 44本目と同じです。それでは実行してみましょう。

…ところが、このクエリを実行しようとすると DBpedia Japaneseのエンドポイントでは、
Virtuoso 42000 Error

The estimated execution time 76656 (sec) exceeds the limit of 500 (sec).

と怒られてしまいます。

推定実行時間が制限 (500秒)を超えてしまうため実行してもらえませんでした。動的に生成される

IRIは 1,620個ほどなので、実行時間の制限を引き上げたエンドポイントを用意すれば、現実的な時間

で結果は得られると思われます。実行計画を分析して、推定実行時間が収まるようにクエリを最適化し

ていく方法もありますが、このノックの範囲を超えるので割愛いたします。

第 5 章

職人級
君はこの魔クエリを受け止められるかな？

これで君も SPARQL職人だ！

実用性度外視のクエリも含む番外編です。

46本目 Virtuoso Serverの情報を取得する

Endpoint OpenLink Virtuoso

1 SELECT
2 (bif:sys_stat(’st_dbms_name’) AS ?name)
3 (bif:sys_stat(’st_dbms_ver’) AS ?version)
4 (bif:sys_stat(’st_build_date’) AS ?date)
5 (bif:sys_stat(’st_build_thread_model’) AS ?thread)
6 (bif:sys_stat(’st_build_opsys_id’) AS ?opsys)
7 WHERE {?s ?p ?o}
8 LIMIT 1

エンドポイントの Virtuoso Serverに関する情報を bif:sys_stat関数*1で取得します。クエリ自体は

全く難しいものではありませんが、ユースケースを考えると職人向けかなと思い、ここで紹介しておき

ました。エンドポイントの外形監視のためにこのクエリを送信して情報取得したり、Virtuosoのバー

ジョンなどによって条件分岐が必要なクエリで使ったりすることが考えられます。

47本目 Fizz Buzz

Endpoint https://ja.dbpedia.org/sparql

1 SELECT ?out
2 WHERE {
3 ?s <http://www.w3.org/2000/01/rdf-schema#label> ?label .
4 BIND (xsd:integer(?label) AS ?num)
5 FILTER (REGEX(?label, "^\\d+$") && ?num > 0)
6 BIND (CONCAT(IF(?num / 3 * 3 = ?num, "Fizz", ""), IF(?num / 5 * 5 = ?num, "Buzz",

"")) AS ?str)
7 BIND (IF(STRLEN(?str) = 0, ?num, ?str) AS ?out)
8 }

*1 http://docs.openlinksw.com/virtuoso/fn_sys_stat/

Interrogating attributes of a Virtuoso Server instance via SPARQL,
http://vos.openlinksw.com/owiki/wiki/VOS/VirtCheckSvrVersionViaSparql

9 GROUP BY ?num
10 ORDER BY ASC(?num)
11 LIMIT 100

Fizz Buzzはご存知の通り、1から順に数を数え、3の倍数なら「Fizz」、5の倍数なら「Buzz」、3と

5の倍数なら「FizzBuzz」と言う言葉遊びです。これをプログラムで実装することで、プログラマを試

験する方法としてもよく知られています。この本を読んだあなたなら、就活の選考で「君、SPARQL

で Fizz Buzz書いてみてよ」と言われても焦らずに済むでしょう！

クエリ自体は大して難しくないのですが、連番を生成する方法が SPARQL標準の機能では思い付か

なかったため (Virtuosoの独自関数を使うと実現できます�48本目)、DBpediaから整数のラベルを

引っ張ってきています。また、SPARQLは剰余演算子がないので、同じ数で割って掛けた値が元の値

と等しいかどうかで判定しています。

48本目 連番を生成する

Endpoint OpenLink Virtuoso

1 SELECT (bif:sequence_next(’my_unique_key’, 1, ?s, ?p, ?o) AS ?id) # 毎回実行
2 ?call_once
3 WHERE {
4 ?s ?p ?o .
5 BIND (bif:sequence_set(’my_unique_key’, 1, 0) AS ?call_once) # 一度だけ実行
6 }
7 LIMIT 100

Virtuoso の独自関数 sequence_set*2、sequence_next*3を使うことで、連番を生成できます。se-

quence_set 関数でシーケンスオブジェクトを初期化し、sequence_next で値を取得&インクリメン

トします。しかし、Virtuosoのオプティマイザは、静的な値しか引数に持たない関数は一度しか実行

せず、すべて同じ値を返すような挙動をします。そのため、オプティマイザをうまく騙す必要があり

ます。

sequence_set 関数は一度実行されればよく、WHERE 句内で実行して変数に結果をバインドし

ておきます（5 行目）。ただし、バインドした変数?call_once をどこかで参照しておかないと最適

化されて sequence_set 関数は一度も実行されないので、SELECT に含めるようにします。次に、

sequence_next 関数は本来 2 つの引数をとり、bif:sequence_next(’my_unique_key’, 1) と書

くことができます。LIMITで指定した 100回実行してほしいのですが、これでは最適化されて一度し

か実行されません。そこで、Virtuosoの独自関数は余分な引数を無視することを利用し、?s、?p、?o

を引数に渡すことで必ず毎回実行されるようにします。

しかしながら、このシーケンスオブジェクトはサーバでグローバルに共有されますし、単一のクエリ

内で連番を生成することを意図した機能ではないと思われます。オプティマイザ回避術もバッドノウ

ハウの塊なので、あまり使おうと思わないほうが良いでしょう…。

*2 http://docs.openlinksw.com/virtuoso/fn_sequence_set/
*3 http://docs.openlinksw.com/virtuoso/fn_sequence_next/

49本目 クエリを実行するクエリ

Endpoint OpenLink Virtuoso

1 SELECT ?sparql ?sql ?exec ?state ?message (bif:length(?rows) AS ?length)
2 (bif:aref(bif:aref(?rows, 0), 0) AS ?concept0) (bif:aref(bif:aref(?rows, 0), 1) AS ?

label0)
3 (bif:aref(bif:aref(?rows, 1), 0) AS ?concept1) (bif:aref(bif:aref(?rows, 1), 1) AS ?

label1)
4 (bif:aref(bif:aref(?rows, 2), 0) AS ?concept2) (bif:aref(bif:aref(?rows, 2), 1) AS ?

label2)
5 WHERE {
6 ?s ?p ?o .
7 BIND (’SELECT * WHERE {?concept a skos:Concept; rdfs:label ?label}’ AS ?sparql)
8 BIND (STR(bif:sparql_to_sql_text(?sparql)) AS ?sql)
9 BIND ("" AS ?state)

10 BIND ("no error" AS ?message)
11 BIND (bif:vector() AS ?meta)
12 BIND (bif:vector() AS ?rows)
13 BIND (bif:exec(?sql, ?state, ?message, bif:vector(), 3, ?meta, ?rows) AS ?exec)
14 } LIMIT 1

クエリ内の文字列リテラルを SPARQLクエリとして評価し、その結果を取得します。Virtuosoの

独自関数を多用していますので、順に説明します。

• 7行目 SPARQLクエリとして実行する文字列を?sparqlにバインド

• 8行目 bif:sparql_to_sql_text関数*4を使い SPARQLを SQLへ変換し?sqlへバインド

(補足) Virtuoso は RDBMS であり RDF データも RDB にストアされています*5。SPARQL

クエリは内部的には SQLに変換することで実行しています。

• 9行目～12行目 bif:exec関数*6に渡す引数のお膳立てです（詳細はドキュメントを参照）。

• 13行目 bif:exec関数で?sqlを実行する。実行結果は?rowsに格納される。

• 2～4行目 検索結果の配列から?conceptと?labelの値を取り出す。

?rowsは Virtuosoの vector型なので、bif:aref*7などの配列操作関数を使う必要があります。

50本目 クワイン (Quine)

Endpoint https://ja.dbpedia.org/sparql

1 SELECT (REPLACE(REPLACE(?s, SUBSTR(?etc, 216, 1), SUBSTR(?etc, 245, 1)), SUBSTR(?etc,
74, 1), ?s) AS ?query) WHERE { <http://ja.dbpedia.org/resource/D.C.P.S.
_～ダ・カーポ～_プラスシチュエーション > <http://ja.dbpedia.org/property/etc> ?etc . }
GROUP BY (’SELECT (REPLACE(REPLACE(?s, SUBSTR(?etc, 216, 1), SUBSTR(?etc, 245, 1)),

*4 http://docs.openlinksw.com/virtuoso/fn_sparql_to_sql_text/
*5 16.1.3. RDF_QUAD and other tables - OpenLink Documentation,

http://docs.openlinksw.com/virtuoso/rdfquadtables/
*6 http://docs.openlinksw.com/virtuoso/fn_exec/
*7 http://docs.openlinksw.com/virtuoso/fn_aref/

SUBSTR(?etc, 74, 1), ?s) AS ?query) WHERE { <http://ja.dbpedia.org/resource/D.C.P.S.
_～ダ・カーポ～_プラスシチュエーション > <http://ja.dbpedia.org/property/etc> ?etc . }
GROUP BY ("&" AS ?s)’ AS ?s)

一見してクエリの整形ミスを疑われるかもしれませんが、そうではないのでご安心を。まず、クワイ

ン (Quine)とは何かご存知でしょうか。簡単に説明すると、クワインとは自身のソースコードと全く

同じ文字列を出力するプログラムのことです。様々なプログラミング言語で一種の娯楽としてクワイ

ンが書かれています。SPARQLのクワインについては、私が Googleなどで検索する限りでは見つか

りませんでしたので、挑戦してみた結果がこのクエリです。なお、SPARQLは問い合わせ言語ですの

で、クエリ文字列とその実行結果であるソリューションの文字列が完全に一致すればクワインであるこ

とにします。

参考として、SQLによるクワインの例を見てみましょう (英語版Wikipediaより*8)。

1 SELECT REPLACE(REPLACE(’SELECT REPLACE(REPLACE("$",CHAR(34),CHAR(39)),CHAR(36),"$") AS
Quine’,CHAR(34),CHAR(39)),CHAR(36),’SELECT REPLACE(REPLACE("$",CHAR(34),CHAR(39)),CHAR
(36),"$") AS Quine’) AS Quine

REPLACEがいくつも書かれていると思いますが、これはリテラルを囲うシングルクォートがリテ

ラル内に出現しないようにしながら、自分自身の文字列となるようにうまく置換をしているのです。ち

なみに、CHAR(34)、CHAR(36)、CHAR(39)はそれぞれダブルクォート (")、ドル記号 ($)、シングル

クォート (’)です。

基本的な戦略はこの SQLのクワインと全く同じです。SPARQLのリテラルはダブルクォートまた

はシングルクォートで囲まれますので、これをいかにうまく置換してやるかが要となります。SQLの

CHAR関数と同等の関数が SPARQLにあるならば簡単なのですが、別の方法でエンコードすること

を考えます。サンプルクエリでは、DBpedia Japaneseからシングル/ダブルクォートを含む適当な文

字列 (?etc) を取ってきて、それを参照して REPLACE することで実現しています。SUBSTR で必

要な文字を取り出していて、SUBSTR(?etc, 216, 1)、SUBSTR(?etc, 245, 1)、SUBSTR(?etc,

74, 1)はそれぞれダブルクォート (")、アンパサンド (&)です。なお、GROUP BYを使っています

が、変数?sへ文字列をバインドするためで、集約としての意味はありません。あとはこれらを使って

うまく REPLACEしていけば完成です。

しかし、このサンプルクエリには重大な欠陥があることを指摘しておかねばなりません。外部リソー

スを参照するという禁じ手を使ってしまっているので厳密にはルール違反です。また、誰かが当該箇所

を書き換えてしまうと、クワインとして成立しなくなるかもしれません。ただ、Web上のリソースを

活用するというのはいかにも SPARQLらしくて良いのではないでしょうか（言い訳）。もっと良いク

ワインを考案された方はぜひ教えてください。

*8 “Quine (computing)”, https://en.wikipedia.org/wiki/Quine_%28computing%29

